```
> restart: with(LinearAlgebra):
```

Aufgabe 36

Die Lösungsmenge L(A,b) eines Systems linearer Gleichungen wird (falls eine Lösung existiert) durch (irgend)eine Lösung dieses Systems und durch (irgend)eine Basis der Lösungsmenge L(A,0) des entsprechenden homogenen Systems linearer Gleichungen beschrieben. Diese Daten können durch den Gauß-Algorithmus berechnet werden: Zuerst wendet man elementare Umformungen auf die erweiterte Matrix (A|b) so lange an, bis PA in der umgeformten Matrix (PA|Pb) Stufenform hat. Dann ist L(A,b) = L(PA,Pb) und die gesuchten Daten können aus (PA|Pb) direkt abgelesen werden.

```
A:=Matrix(4,6,[0,2,2,-3,2,1,-2,2,4,-3,1,0,1,1,-1,0,-2,1,0,1,2,3,4,
1]);
```

$$A := \begin{bmatrix} 0 & 2 & 2 & -3 & 2 & 1 \\ -2 & 2 & 4 & -3 & 1 & 0 \\ 1 & 1 & -1 & 0 & -2 & 1 \\ 0 & 1 & 2 & 3 & 4 & 1 \end{bmatrix}$$

```
> b:=Vector(<2,0,1,1>);
```

$$b := \begin{bmatrix} 2 \\ 0 \\ 1 \\ 1 \end{bmatrix}$$

```
> c:=Vector(<-1,1,0,2>);
```

$$c := \begin{bmatrix} -1 \\ 1 \\ 0 \\ 2 \end{bmatrix}$$

```
> RedeucedRowEchelonForm(<A|b>);
```

```
> ReducedRowEchelonForm(<A|b>);
```
Die erweiterte Matrix (A|b) wird elementar auf (PA|Pb) umgeformt.

$$\begin{bmatrix} 1 & 0 & 0 & 0 & \frac{31}{8} & \frac{5}{8} & \frac{7}{4} \\ 0 & 1 & 0 & 0 & \frac{-5}{2} & \frac{1}{2} & 0 \\ 0 & 0 & 1 & 0 & \frac{27}{8} & \frac{1}{8} & \frac{3}{4} \\ 0 & 0 & 0 & 1 & \frac{-1}{12} & \frac{1}{12} & \frac{-1}{6} \end{bmatrix}$$

Die folgende Spalte ist eine Lösung des Gleichungssystems Ax=b :

```
> zb:=Vector(<7/4, 0, 3/4, -1/6, 0,0>);
```

$$zb := \begin{bmatrix} \frac{7}{4} \\ 0 \\ \frac{3}{4} \\ \frac{-1}{6} \\ 0 \\ 0 \end{bmatrix}$$

```
> ReducedRowEchelonForm(<A|c>);
```

Die erweiterte Matrix (A|c) wird elementar auf (PA|Pc) umgeformt.

$$\begin{bmatrix} 1 & 0 & 0 & 0 & \frac{31}{8} & \frac{5}{8} & \frac{-3}{2} \\ 0 & 1 & 0 & 0 & \frac{-5}{2} & \frac{1}{2} & 1 \\ 0 & 0 & 1 & 0 & \frac{27}{8} & \frac{1}{8} & \frac{-1}{2} \\ 0 & 0 & 0 & 1 & \frac{-1}{12} & \frac{1}{12} & \frac{2}{3} \end{bmatrix}$$

\> Die folgende Spalte ist eine Lösung des Gleichungssystems Ax=c :

```
> zc:=Vector(<-3/2, 1, -1/2, 2/3, 0,0>);
```

$$zc := \begin{bmatrix} \frac{-3}{2} \\ 1 \\ \frac{-1}{2} \\ \frac{2}{3} \\ 0 \\ 0 \end{bmatrix}$$

```
> v:=NullSpace(A);
```

Das folgende Paar von Spalten ist eine Basis von L(A,0).

$$v := \left\{ \begin{bmatrix} \frac{-5}{8} \\ \frac{-1}{2} \\ \frac{-1}{8} \\ \frac{-1}{12} \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} \frac{-31}{8} \\ \frac{5}{2} \\ \frac{-27}{8} \\ \frac{1}{12} \\ 1 \\ 0 \end{bmatrix} \right\}$$

Die Lösungsmenge L(A,b) bzw. L(A,c) ist die Menge { zb + d_1.v[1] + d_2.v[2] | d_1, d_2   in **Q** (oder **R** oder **C**) } bzw.
{ zc + d_1.v[1] + d_2.v[2] | d_1, d_2 in **Q** (oder **R** oder **C**) } .

```
> restart: with(LinearAlgebra):
```

Aufgabe 37

Zur Lösung dieser "Textaufgabe" überlegen wir zuerst, was gesucht ist:
Gesucht sind vier Zahlen, nämlich die Anzahlen der Bienen am Anfang in den Räumen A, B, C, D. Wir bezeichnen diese Zahlen mit a, b, c, d.
Dann lesen wir aus dem Text die Bedingungen ab, die diese vier Zahlen erfüllen müssen: Im Raum A sind nach einer Minute 170 Bienen. Woher kommen diese? Im Raum A bleiben 40/100.a Bienen, aus den Räumen B, C, D kommen 30/100.b, 25/100.c und 35/100.d Bienen. Also muss 1/100.(40a + 30b+25c+35d) = 170 sein. Für die Räume B, C, D erhalten wir auf ähnliche Weise je eine Bedingung für die vier Zahlen a, b, c, d.
In Matrizenform bedeutet das: Die Spalte

```
> u:=Vector(<a,b,c,d>);
```

$$u := \begin{bmatrix} a \\ b \\ c \\ d \end{bmatrix}$$

ist Lösung des Systems linearer Gleichungen Ax=b, wobei

```
> A:=(1/100)*Matrix(4,4,[40, 30, 25, 35, 20, 40, 25, 0, 20, 30, 50, 0, 20, 0, 0, 65]);  b:=Vector(<170, 104, 112, 114>);
  Eigenvalues(A);
```

$$A := \begin{bmatrix} \frac{2}{5} & \frac{3}{10} & \frac{1}{4} & \frac{7}{20} \\ \frac{1}{5} & \frac{2}{5} & \frac{1}{4} & 0 \\ \frac{1}{5} & \frac{3}{10} & \frac{1}{2} & 0 \\ \frac{1}{5} & 0 & 0 & \frac{13}{20} \end{bmatrix}$$

$$b := \begin{bmatrix} 170 \\ 104 \\ 112 \\ 114 \end{bmatrix}$$

Elementare Umformungen von (A|b) ergeben

```
> ReducedRowEchelonForm(<A|b>);
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 & 180 \\ 0 & 1 & 0 & 0 & 120 \\ 0 & 0 & 1 & 0 & 80 \\ 0 & 0 & 0 & 1 & 120 \end{bmatrix}$$

Also ist die gesuchte Spalte u gleich

```
> u:=Vector(<180, 120, 80,120>);
```

$$u := \begin{bmatrix} 180 \\ 120 \\ 80 \\ 120 \end{bmatrix}$$

Die Anzahlen der Bienen in den vier Räumen werden nach einer Minute bzw. nach zwei Minuten bzw. nach drei Minuten durch A.u bzw. A.(A.u) = A^2.u bzw. A.(A^2.u) = A^3.u beschrieben. Nach 1000 Minuten durch A^1000.u . Mit dem Befehl "evalf" werden die Einträge in den Spalten durch Dezimalzahlen (näherungsweise) dargestellt.

```
> evalf(A.u); evalf((A^2).u); evalf((A^3).u); evalf((A^1000).u);
```

$$\begin{bmatrix} 170. \\ 104. \\ 112. \\ 114. \end{bmatrix}$$

$$\begin{bmatrix} 167.1000000 \\ 103.6000000 \\ 121.2000000 \\ 108.1000000 \end{bmatrix}$$

$$\begin{bmatrix} 166.0550000 \\ 105.1600000 \\ 125.1000000 \\ 103.6850000 \end{bmatrix}$$

$$\begin{bmatrix} 164.5768025 \\ 109.7178683 \\ 131.6614420 \\ 94.04388715 \end{bmatrix}$$

```
> restart: with(LinearAlgebra):
```

Mit "restart" werden die bisher getroffenen Vereinbarungen (insbesondere Bezeichnungen) gelöscht, wir müssen dann allerdings das Paket "LinearAlgebra" auch neu laden.

Aufgabe 38

Die zu einer Matrix A inverse Matrix B ist eine Matrix mit der Eigenschaft A.B = B.A = I_n (Einheitsmatrix). Wenn eine solche Matrix existiert, heißt A invertierbar.
Wenn A eine n x n - Matrix ist, dann kann auf die folgende Weise überprüft werden, ob A invertierbar ist und gegebenenfalls die zu A inverse Matrix A^(-1) berechnet werden: Forme die Matrix (A|I_n) so lange zu (P.A|P.I_n) um, bis P.A Stufenform hat. Wenn P.A dann die

Einheitsmatrix ist, dann ist A invertierbar, sonst nicht. Wenn A invertierbar ist, dann ist P die zu A inverse Matrix.

```
> A:= Matrix(3,[2,0,1,3,1,4,-2,1,2]);
```

$$A := \begin{bmatrix} 2 & 0 & 1 \\ 3 & 1 & 4 \\ -2 & 1 & 2 \end{bmatrix}$$

```
> ReducedRowEchelonForm(A);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

Die Matrix A ist also invertierbar! Wir berechnen durch Umformen von (A| I_3) die zu ihr inverse Matrix:

```
> ReducedRowEchelonForm(<A|IdentityMatrix(3)>);
```

$$\begin{bmatrix} 1 & 0 & 0 & -2 & 1 & -1 \\ 0 & 1 & 0 & -14 & 6 & -5 \\ 0 & 0 & 1 & 5 & -2 & 2 \end{bmatrix}$$

Man kann die zu A inverse Matrix auch direkt durch die Eingabe von A^(-1) berechnen:

```
> A^(-1);
```

$$\begin{bmatrix} -2 & 1 & -1 \\ -14 & 6 & -5 \\ 5 & -2 & 2 \end{bmatrix}$$

```
> B:=Matrix(4,[2,1,3,4,1,4,0,2,1,-1,2,2,2,6,1,4]);
```

$$B := \begin{bmatrix} 2 & 1 & 3 & 4 \\ 1 & 4 & 0 & 2 \\ 1 & -1 & 2 & 2 \\ 2 & 6 & 1 & 4 \end{bmatrix}$$

```
> B^(-1);
Error, (in rtable/Power) singular matrix
```

B ist nicht invertierbar! Wir können das überprüfen, indem wir die Stufenform von B berechnen:

```
> ReducedRowEchelonForm(B);
```

$$\begin{bmatrix} 1 & 0 & 0 & 2 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

Aufgabe 39

```
> restart: with(LinearAlgebra):
```

```
> A:= Matrix([[-6,2,-12, 6],[-4,a,-11,0],[4,-6,1,b]]);
```

$$A := \begin{bmatrix} -6 & 2 & -12 & 6 \\ -4 & a & -11 & 0 \\ 4 & -6 & 1 & b \end{bmatrix}$$

Wir formen die Matrix schrittweise in Richtung Stufenform um. Wir formen aber nur so lange um, bis wir wissen, wie viele Pivots die Matrix hat.

```
> A1:=Pivot(A,1,1);
```
Zur zweiten und zur dritten Zeile werden geeignete Vielfache der ersten Zeile addiert, damit in diesen Zeilen in der ersten Spalte Nullen stehen.

$$A1 := \begin{bmatrix} -6 & 2 & -12 & 6 \\ 0 & a-\frac{4}{3} & -3 & -4 \\ 0 & \frac{-14}{3} & -7 & b+4 \end{bmatrix}$$

```
> A2:=RowOperation(A1, [2,3]);
```
Die zweite und die dritte Zeile werden vertauscht. Damit wird vermieden, im nächsten Schritt die Fälle "a = 4/3" und "a ist nicht 4/3" unterscheiden zu müssen.

$$A2 := \begin{bmatrix} -6 & 2 & -12 & 6 \\ 0 & \frac{-14}{3} & -7 & b+4 \\ 0 & a-\frac{4}{3} & -3 & -4 \end{bmatrix}$$

```
> A3:=RowOperation(A2,[3,2], (a-4/3)*(3/14));
```
Zur dritten Zeile addieren wir das (a-4/3).(-3/14) - fache der zweiten Zeile.

$$A3 := \begin{bmatrix} -6 & 2 & -12 & 6 \\ 0 & \frac{-14}{3} & -7 & b+4 \\ 0 & 0 & -1-\frac{3\,a}{2} & -4+\left(\frac{3\,a}{14}-\frac{2}{7}\right)(b+4) \end{bmatrix}$$

Wenn -1-(3a/2) nicht 0 ist, also a nicht -2/3 ist, dann hat die Stufenform der Matrix A3 (und damit auch die Stufenform von A) in jeder Zeile und jeder Spalte einen Pivot. Also hat in diesem Fall das Gleichungssystem genau eine Lösung (für jede Zahl b).
Wenn a = -2/3 ist, dann gibt es keine Lösung, wenn der Eintrag von A3 in der 3. Zeile und 4. Spalte nicht 0 ist, und unendlich viele Lösungen, wenn dieser Eintrag gleich 0 ist. Eine einfache Rechnung ergibt: dieser Eintrag ist genau dann 0, wenn b = -40/3 ist.

```
> restart: with(LinearAlgebra):
```

Aufgabe 40

Wir bezeichnen mit a, b, c, d, e, f, g die Temperaturen in den Gitterpunkten A, B, C, D, E, F, G.
Dann ist 3a-b-c=g, 2b-a-d = 0, 2c-a-d=0, 3d-b-c-e=0 und 2e-d=f.

```
> restart: with(LinearAlgebra): A:=Matrix(5,[3,-1,-1,0,0,
  -1,2,0,-1,0, -1,0,2,-1,0, 0,-1,-1,3,-1, 0,0,0,-1,2]);
```

$$A := \begin{bmatrix} 3 & -1 & -1 & 0 & 0 \\ -1 & 2 & 0 & -1 & 0 \\ -1 & 0 & 2 & -1 & 0 \\ 0 & -1 & -1 & 3 & -1 \\ 0 & 0 & 0 & -1 & 2 \end{bmatrix}$$

```
> r:=Vector(<g,0,0,0,f>);
```

$$r := \begin{bmatrix} g \\ 0 \\ 0 \\ 0 \\ f \end{bmatrix}$$

```
> ReducedRowEchelonForm(<A|r>);
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 & 0 & \frac{3g}{4} + \frac{f}{4} \\ 0 & 1 & 0 & 0 & 0 & \frac{5g}{8} + \frac{3f}{8} \\ 0 & 0 & 1 & 0 & 0 & \frac{5g}{8} + \frac{3f}{8} \\ 0 & 0 & 0 & 1 & 0 & \frac{g}{2} + \frac{f}{2} \\ 0 & 0 & 0 & 0 & 1 & \frac{3f}{4} + \frac{g}{4} \end{bmatrix}$$

Die Temperaturen a, ..., e stehen in der 6. Spalte dieser Matrix. Die Temperatur in C ist genau dann 0, wenn 5g+3f=0 ist, also wenn g = -3/5.f ist.

Aufgabe 41

Ein n-Tupel (v_1, ... , v_n) von Vektoren eines Vektorraums V (über einem Körper K) ist eine Basis von V genau dann, wenn jeder Vektor in V auf genau eine Weise als Linearkombination (mit Koeffizienten in K) von (v_1, ... , v_n) geschrieben werden kann.

Zu jedem Vektor w in V gibt es dann eindeutig bestimmte Zahlen (Elemente von K) c_1, ... , c_n so, dass

w = c_1.v_1 + ... + c_n.v_n ist.

Diese Zahlen heißen dann "Koordinaten von w bezüglich der Basis (v_1, ... , v_n)".

Die Spalten A_(-1), A_(-2), A_(-3) von A bilden genau dann eine Basis von **Q**^(3 x 1), wenn es zu jeder Spalte b
in **Q**^(3 x 1) eindeutig bestimmte rationale
Zahlen x_1, x_2, x_3 gibt so, dass x_1.A_(-1)+x_2.A_(-2)+x_3. A_(-3) = b ist. Das ist aber genau dann der Fall, wenn es zu jeder Spalte b genau eine Spalte x mit A.x = b gibt. Das wiederum ist genau dann der Fall, wenn die Stufenform von A die Einheitsmatrix ist. Die Koordinatenspalte von

b bezüglich der Basis (A_(-1), A_(-2), A_(-3)) ist dann die eindeutig bestimmte Lösung von A.x = b, also A^(-1).b .

```
> A:= Matrix(3, [1,2,0,1,2,1,-1,4,2]);
```

$$A := \begin{bmatrix} 1 & 2 & 0 \\ 1 & 2 & 1 \\ -1 & 4 & 2 \end{bmatrix}$$

```
> ReducedRowEchelonForm(A);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

Die Spalten von A bilden also eine Basis von **Q**^(3 x 1).

```
> u:= Vector(<1,2,3>);
```

$$u := \begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix}$$

```
> v:=Vector(<1,0,0>);
```

$$v := \begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix}$$

```
> A^(-1).u;
```

Wir erhalten die Koordinatenspalte von u bezüglich der Basis (A_(-1), A_(-2), A_(-3)) .

$$\begin{bmatrix} \frac{1}{3} \\ \frac{1}{3} \\ 1 \end{bmatrix}$$

```
> A^(-1).v;
```

Wir erhalten die Koordinatenspalte von u bezüglich der Basis (A_(-1), A_(-2), A_(-3)) .

$$\begin{bmatrix} 0 \\ \frac{1}{2} \\ -1 \end{bmatrix}$$

```
> ReducedRowEchelonForm(<A|u|v>);
```

Eine andere Möglichkeit, die Koordinatenspalten von u und von v (also die Lösungen von A.x=u und A.x=v) zu berechnen.

$$\begin{bmatrix} 1 & 0 & 0 & \frac{1}{3} & 0 \\ 0 & 1 & 0 & \frac{1}{3} & \frac{1}{2} \\ 0 & 0 & 1 & 1 & -1 \end{bmatrix}$$

```
>
```